# NAG Library Routine Document

# F07PSF (ZHPTRS)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F07PSF (ZHPTRS) solves a complex Hermitian indefinite system of linear equations with multiple right-hand sides,

$$AX = B,$$

where $A$ has been factorized by F07PRF (ZHPTRF), using packed storage.

## 2 Specification

```
SUBROUTINE F07PSF (UPLO, N, NRHS, AP, IPIV, B, LDB, INFO)

INTEGER               N, NRHS, IPIV(*), LDB, INFO
COMPLEX (KIND=nag_wp) AP(*), B(LDB,*)
CHARACTER(1)          UPLO
```

The routine may be called by its LAPACK name ***zhptrs***.

## 3 Description

F07PSF (ZHPTRS) is used to solve a complex Hermitian indefinite system of linear equations $AX = B$, the routine must be preceded by a call to F07PRF (ZHPTRF) which computes the Bunch–Kaufman factorization of $A$, using packed storage.

If UPLO = 'U', $A = PUDU^{\mathrm{H}}P^{\mathrm{T}}$, where $P$ is a permutation matrix, $U$ is an upper triangular matrix and $D$ is an Hermitian block diagonal matrix with 1 by 1 and 2 by 2 blocks; the solution $X$ is computed by solving $PUDY = B$ and then $U^{\mathrm{H}}P^{\mathrm{T}}X = Y$.

If UPLO = 'L', $A = PLDL^{\mathrm{H}}P^{\mathrm{T}}$, where $L$ is a lower triangular matrix; the solution $X$ is computed by solving $PLDY = B$ and then $L^{\mathrm{H}}P^{\mathrm{T}}X = Y$.

## 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

## 5 Parameters

1: UPLO – CHARACTER(1) *Input*

*On entry*: specifies how $A$ has been factorized.

UPLO = 'U'
: $A = PUDU^{\mathrm{H}}P^{\mathrm{T}}$, where $U$ is upper triangular.

UPLO = 'L'
: $A = PLDL^{\mathrm{H}}P^{\mathrm{T}}$, where $L$ is lower triangular.

*Constraint*: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix $A$.

*Constraint*: $\mathrm{N} \geq 0$.

3: NRHS – INTEGER *Input*

*On entry*: $r$, the number of right-hand sides.

*Constraint*: $\mathrm{NRHS} \geq 0$.

4: AP($*$) – COMPLEX (KIND=nag_wp) array *Input*

**Note**: the dimension of the array AP must be at least $\max(1, \mathrm{N} \times (\mathrm{N}+1)/2)$.

*On entry*: the factorization of $A$ stored in packed form, as returned by F07PRF (ZHPTRF).

5: IPIV($*$) – INTEGER array *Input*

**Note**: the dimension of the array IPIV must be at least $\max(1, \mathrm{N})$.

*On entry*: details of the interchanges and the block structure of $D$, as returned by F07PRF (ZHPTRF).

6: B(LDB,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

**Note**: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

*On entry*: the $n$ by $r$ right-hand side matrix $B$.

*On exit*: the $n$ by $r$ solution matrix $X$.

7: LDB – INTEGER *Input*

*On entry*: the first dimension of the array B as declared in the (sub)program from which F07PSF (ZHPTRS) is called.

*Constraint*: $\mathrm{LDB} \geq \max(1, \mathrm{N})$.

8: INFO – INTEGER *Output*

*On exit*: $\mathrm{INFO} = 0$ unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\mathrm{INFO} < 0$

If $\mathrm{INFO} = -i$, the $i$th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

For each right-hand side vector $b$, the computed solution $x$ is the exact solution of a perturbed system of equations $(A+E)x = b$, where

if UPLO = 'U', $|E| \leq c(n)\epsilon P|U||D||U^{\mathrm{H}}|P^{\mathrm{T}}$;

if UPLO = 'L', $|E| \leq c(n)\epsilon P|L||D||L^{\mathrm{H}}|P^{\mathrm{T}}$,

$c(n)$ is a modest linear function of $n$, and $\epsilon$ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution $x$ satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq c(n)\,\mathrm{cond}(A,x)\epsilon$$

where $\mathrm{cond}(A,x) = \left\||A^{-1}||A||x|\right\|_\infty / \|x\|_\infty \leq \mathrm{cond}(A) = \left\||A^{-1}||A|\right\|_\infty \leq \kappa_\infty(A)$.

Note that $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$.

Forward and backward error bounds can be computed by calling F07PVF (ZHPRFS), and an estimate for $\kappa_\infty(A)$ $(= \kappa_1(A))$ can be obtained by calling F07PUF (ZHPCON).

# 8 Further Comments

The total number of real floating point operations is approximately $8n^2r$.

This routine may be followed by a call to F07PVF (ZHPRFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07PEF (DSPTRS).

# 9 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} -1.36+0.00i & 1.58+0.90i & 2.21-0.21i & 3.91+1.50i \\ 1.58-0.90i & -8.87+0.00i & -1.84-0.03i & -1.78+1.18i \\ 2.21+0.21i & -1.84+0.03i & -4.63+0.00i & 0.11+0.11i \\ 3.91-1.50i & -1.78-1.18i & 0.11-0.11i & -1.84+0.00i \end{pmatrix}$$

and

$$B = \begin{pmatrix} 7.79+\ \ 5.48i & -35.39+18.01i \\ -0.77-16.05i & 4.23-70.02i \\ -9.58+\ \ 3.88i & -24.79-\ \ 8.40i \\ 2.98-10.18i & 28.68-39.89i \end{pmatrix}.$$

Here $A$ is Hermitian indefinite, stored in packed form, and must first be factorized by F07PRF (ZHPTRF).

## 9.1 Program Text

```
    Program f07psfe

!     F07PSF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zhptrf, zhptrs
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, j, ldb, n, nrhs
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ap(:), b(:,:)
      Integer, Allocatable             :: ipiv(:)
      Character (1)                    :: clabs(1), rlabs(1)
!     .. Executable Statements ..
      Write (nout,*) 'F07PSF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      ldb = n
```

```
      Allocate (ap(n*(n+1)/2),b(ldb,nrhs),ipiv(n))

!     Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!     Factorize A
!     The NAG name equivalent of zhptrf is f07prf
      Call zhptrf(uplo,n,ap,ipiv,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Compute solution
!       The NAG name equivalent of zhptrs is f07psf
        Call zhptrs(uplo,n,nrhs,ap,ipiv,b,ldb,info)

!       Print solution

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed','F7.4', &
          'Solution(s)','Integer',rlabs,'Integer',clabs,80,0,ifail)

      Else
        Write (nout,*) 'The factor D is singular'
      End If

    End Program f07psfe
```

## 9.2 Program Data

```
F07PSF Example Program Data
  4  2                                                  :Values of N and NRHS
  'L'                                                   :Value of UPLO
 (-1.36, 0.00)
 ( 1.58,-0.90) (-8.87, 0.00)
 ( 2.21, 0.21) (-1.84, 0.03) (-4.63, 0.00)
 ( 3.91,-1.50) (-1.78,-1.18) ( 0.11,-0.11) (-1.84, 0.00)  :End of matrix A
 ( 7.79,  5.48) (-35.39, 18.01)
 (-0.77,-16.05) (  4.23,-70.02)
 (-9.58,  3.88) (-24.79, -8.40)
 ( 2.98,-10.18) ( 28.68,-39.89)                         :End of matrix B
```

## 9.3 Program Results

```
 F07PSF Example Program Results

 Solution(s)
                    1                  2
 1  ( 1.0000,-1.0000) ( 3.0000,-4.0000)
 2  (-1.0000, 2.0000) (-1.0000, 5.0000)
 3  ( 3.0000,-2.0000) ( 7.0000,-2.0000)
 4  ( 2.0000, 1.0000) (-8.0000, 6.0000)
```